

```

!
! Test program for TAMS 1805A switch controller
! (c) Tech Soft GmbH, Sven Henze
!
! Created: 03-May-2006
! Update: 13-Oct-2006
!
! This program is "hammering" the TAMS 1805A switch controller
!
DIM Cmd$[80],A$[1024],Devname$[255]
INTEGER Numloops,Numswitches,Verbose
LONG Cyc
!
Verbose=0
!
CONTROL KBD,203;1
CONTROL KBD,204;1
!
RANDOMIZE
!
CONTROL CRT,100;1
!
G$=CHR$(19) ! character code for inverse printing
X$=CHR$(16) ! character code for normal printing
Hl$=CHR$(30) ! character code for "highlighted" printing
Hx$=CHR$(24) ! character code for "normal" printing
Ul$=CHR$(22) ! character code for underlined printing
CrLf$=CHR$(10)&CHR$(13)
!
IF FNHtbvers>=9.4 THEN ! only possible with HTBasic 9.4 and higher
    Devname$="UsbDevice1"
    ON ERROR GOTO Drv_load_err
    !
    IF SYSTEM$("VERSION:USBS")="0" THEN LOAD BIN "USBS;DEV "&Devname$&" TIME 10"
    CLEAR SCREEN
    !
    PRINT G$&" TAMS 1805A test program "&X$
    PRINT
    !
    Dummy=FNSend_command(7,"*IDN?") ! returns identifier
    PRINT Ul$&"Device ID"&X$&": ";FNRead_device$(7)
    !
    Dummy=FNSend_command(7,"*RST") ! reset device
    !
    Dummy=FNSend_command(7,"*TST?")
    PRINT Ul$&"Test result"&X$&": ";FNRead_device$(7)
    !
    STATUS 7,1;Reg1 ! does device implement the USBTMC protocol?
    STATUS 7,2;Reg2 ! is TRIGGER supported?
    STATUS 7,3;Reg3 ! is REMOTE supported?
    STATUS 7,4;Reg4 ! are SCPI commands supported?
    PRINT Ul$&"Register 1..4"&X$&": ";Reg1;" ";Reg2;" ";
    PRINT Reg3;" ";Reg4;" (USBTMC, Trigger, Remote, SCPI)"
    PRINT
    !
    Numloops=5
    Numswitches=5 ! the 1805A has 5 main relays (switches)
    !
    REAL Zeit=.05
    Randomtime=0
    !
    Cyc=0
    R$=VAL$(PROUND(Zeit,-3))&" s "
    IF R$[1;1]="." THEN R$="0"&R$
    IF Randomtime THEN R$="(random) "
    PRINT Ul$&"Switch delay"&X$&": ";R$
    PRINT
    PRINT Ul$&"Switch command utilization"&X$
    Sl=10
    !

```

```

User_break=0
ON KBD ALL GOSUB Kbd_break
DISP "Press any key to stop the program"
!
Faktor=10
WHILE NOT User_break
!
FOR I=1 TO Numloops
IF User_break THEN I=Numloops
Cyc=Cyc+1
!
FOR J=1 TO Numswitches
IF Randomtime THEN Zeit=MAX(.01,RND/Faktor)
IF User_break THEN J=Numswitches
Cmd$="CLOSE (@"&VAL$(J)&")"
Dummy=FNSend_command(7,Cmd$,Verbose)
PRINT TABXY(2,S1+J);"CLOSE (@";VAL$(J);):";Cyc;" "
WAIT Zeit
Cmd$="OPEN (@"&VAL$(J)&")"
Dummy=FNSend_command(7,Cmd$,Verbose)
PRINT TABXY(22,S1+J);"OPEN (@";VAL$(J);):";Cyc;" "
IF Randomtime THEN Zeit=MAX(.01,RND/Faktor)
WAIT Zeit
NEXT J
NEXT I
END WHILE
ELSE
BEEP
PRINT
ALPHA PEN 2
PRINT " ** ERROR: Wrong HTBasic version **"
ALPHA PEN 1
PRINT " Sorry, you are using HTBasic version "&VAL$(FNHtbvers);
PRINT " which is not compatible with this demo program."
PRINT " This example only works with HTBasic 9.4 and above."
PRINT
PRINT " Program stopped."
STOP
END IF
!
STOP
!
Kbd_break: !
BEEP
User_break=1
DISP "User break, program stopped."
RETURN
!
Drv_load_err: ! error during driver loading
SELECT ERRN
CASE 167 ! Interface Status Error
BEEP
ALPHA PEN 2
PRINT "*** ERROR 167 while loading USB driver"
ALPHA PEN 1
PRINT " - Is the device connected and blinking?"
PRINT " - Is the devicename configured correctly ('&Devname$&')?"
CASE 731
BEEP
ALPHA PEN 2
PRINT "*** ERROR 731 while talking to the SICL driver"
ALPHA PEN 1
PRINT " - Is the device still connected and blinking?"
END SELECT
PRINT
PRINT "Program stopped."
STOP
!
END
!
```

```

DEF FNHtbvers
!
  REAL Htbvers
  DIM Htbvers$(80)
  !
  Htbvers$=SYSTEM$("VERSION:HTB")
  !
  ! version number is located 16 characters after the occurrence of "Release"
  Htbvers=VAL(Htbvers$[(POS(Htbvers$,"Release"))+8;3])
  !
  RETURN Htbvers
!
FNEND
!
DEF FNSend_command(INTEGER Isc,Cmd$,OPTIONAL INTEGER Verbose)
!
  Verb_=0
  IF NPAR>=3 THEN
    Verb_=Verbose
  END IF
  !
  ON TIMEOUT Isc,5 GOTO Sc_timeout
  !
  ! make sure END is specified, otherwise the system may hang
  OUTPUT Isc;Cmd$ END
  IF Verb_ THEN PRINT Cmd$&" ",
  !
  RETURN 1
!
Sc_timeout:      !
  BEEP
  PRINT "*** Timeout ERROR in FNSend_command ***"
  RETURN 0
FNEND
!
DEF FNRead_device$(INTEGER Isc)
  DIM A$[1024]
  !
  ON TIMEOUT Isc,5 GOTO Rd_timeout
  !
  ENTER Isc;A$
  !
  RETURN A$
!
Rd_timeout:    !
  BEEP
  PRINT "*** Timeout ERROR in FNRead_device ***"
  RETURN ""
FNEND

```